

Vážení zákazníci,

dovolujeme si Vás upozornit, že na tuto ukázkou knihy se vztahují autorská práva, tzv. copyright.


To znamená, že ukáзка má sloužit výhradně pro osobní potřebu potenciálního kupujícího (aby čtenář viděl, jakým způsobem je titul zpracován a mohl se také podle tohoto, jako jednoho z parametrů, rozhodnout, zda titul koupí či ne).

Z toho vyplývá, že není dovoleno tuto ukázkou jakýmkoliv způsobem dále šířit, veřejně či neveřejně např. umístováním na datová média, na jiné internetové stránky (ani prostřednictvím odkazů) apod.

redakce nakladatelství BEN – technická literatura
redakce@ben.cz



Výběr a kopírování



Obr. 50 **G14.DPR**

Co to dělá:

Stiskem myši začíná výběr obdélníka v Image1 (levý), při pohybu myši se mění jeho rozměry a po puštění tlačítka je vybraná část obrázku překopírována do levého horního rohu obrázku Image2.

Jak:

```
var
  x1,y1: integer; { prvni roh }
  sx,sy: integer; { druhy roh }
  dole: boolean;
  ...
procedure TForm1.Image1MouseDown(Sender: TObject;
  Button: TMouseButton;
  Shift: TShiftState; X, Y: Integer);
begin
  x1 := X;
  y1 := Y;
  dole := TRUE;

  Image1.Canvas.Brush.Style := bsClear;
  Image1.Canvas.Pen.Style := psDash;
```

```

    Image1.Canvas.Pen.Mode := pmNot;

    sx := X; sy := Y;

    { namalovat nový obdelnik: }
    Image1.Canvas.Rectangle( x1,y1,sx,sy );
end;

procedure TForm1.Image1MouseMove(Sender: TObject;
    Shift: TShiftState; X, Y: Integer);
begin
    if dole then
    begin
        { vymazat minuly obdelnik: }
        Image1.Canvas.Rectangle( x1,y1,sx,sy );

        sx := X; sy := Y;

        { namalovat nový obdelnik: }
        Image1.Canvas.Rectangle( x1,y1,sx,sy );
    end
end;

procedure TForm1.Image1MouseUp(Sender: TObject;
    Button: TMouseButton;
    Shift: TShiftState; X, Y: Integer);
begin
    dole := FALSE;

    { vymazat minuly obdelnik: }
    Image1.Canvas.Rectangle( x1,y1,sx,sy );

    Image2.Canvas.CopyRect (
        Rect( 0,0,X-x1+1,Y-y1+1 ),
        Image1.Canvas,
        Rect( x1,y1,X,Y ) )
end;

```

Vysvětlení

Metoda **CopyRect** objektu Canvas kopíruje obdélník ze stejného nebo jiného Canvasu. Parametry udávají **cílový obdélník**, **zdrojový Canvas** (cílovým Canvasem je Canvas, jehož metodu CopyRect voláme) a **zdrojový obdélník**.

Při různé velikosti zdrojového a cílového obdélníka dojde k příslušné deformaci.

Poznámka

Rect je funkce, která ze čtyř souřadnic vyrobí hodnotu typu TRect (!! funkce vracející record!! v Pascalu!!) – obdélník.

Pozor

Existují objekty Canvas, ze kterých nelze číst (Printer.Canvas).

Poznámka

Ano, tisknout z Delphi je snadné – objekt **Printer** (unita *Printers*) má vlastnost *Canvas* odpovídající právě tištěné stránce.

Poznámka


Tedy ten tisk – třeba takhle:

```
...
Printer.BeginDoc;
Printer.Canvas.MoveTo( 100,100 );
Printer.Canvas.LineTo( 200,200 );
Printer.EndDoc
...
```

Poznámka

Souřadnice myši **X** a **Y** v parametrech procedur **<nějaký objekt>MouseDown** atd. jsou **relativní** vzhledem k objektu **Sender**.

Skládačka



Obr. 51 G15.DPR

Co to dělá:

Načte vybraný obrázek (vybraný pomocí `OpenDialog1`), vytvoří $M \times N$ komponent `TImage` obsahujících odpovídající části načteného obrázku a tyto části dovolí přesouvat myší (`MouseDown`, `MouseMove`, `MouseUp`). Při konci přesouvání tyto komponenty doskakují na pozice odpovídající mřížce zadaných rozměrů.

Jak:

```
var
  sx,sy: integer;
  dole: boolean;

procedure TForm1.Image1MouseDown(Sender: TObject;
  Button: TMouseButton;
  Shift: TShiftState; X, Y: Integer);
begin
  sx := X; sy := Y;
  dole := TRUE
end;
```

```

var vx,vy: integer;

procedure TForm1.Image1MouseMove(Sender: TObject;
    Shift: TShiftState; X, Y: Integer);
begin
    if dole then
        with (Sender as TImage) do
            begin
                Left := Left + X-sx;
                Top := Top + Y-sy
            end
        end;
end;

procedure TForm1.Image1MouseUp(Sender: TObject;
    Button: TMouseButton;
    Shift: TShiftState; X, Y: Integer);
begin
    with (Sender as TImage) do
        begin
            Left := vx * round( Left / vx );
            Top := vy * round( Top / vy )
        end;
        dole := FALSE
    end;
end;

procedure TForm1.FormClick(Sender: TObject);
var PX,PY: integer;
    BM: TBitmap;
    x,y: integer;
    I: TImage;
begin
    if not OpenFileDialog1.Execute then exit;

    try
        PX := StrToInt( Edit1.Text );
        PY := StrToInt( Edit2.Text )
    except
        ShowMessage( 'Špatně zadané rozměry' );
        exit
    end;

    BM := TBitmap.Create;
    BM.LoadFromFile( OpenFileDialog1.FileName );

    vx := (BM.Width div PX);
    vy := (BM.Height div PY);

```

```

for x:=1 to PX do
for y:=1 to PY do
begin
  I := TImage.Create( Self );
  I.Left := (x-1)*vx;
  I.Top := (y-1)*vy;
  I.Width := vx;
  I.Height := vy;

  I.Canvas.CopyRect( Rect( 0,0, vx-1,vy-1 ),
                    BM.Canvas,
                    Rect( (x-1)*vx,
                        (y-1)*vy,
                        x*vix,
                        y*vy ) );

  I.Parent := Self;
  I.OnMouseDown := Image1MouseDown;
  I.OnMouseMove := Image1MouseMove;
  I.OnMouseUp := Image1MouseUp
end;

BM.Free

end;

```

Vysvětlení

Snad není třeba.